ABSTRACT

In clinical trials data analysis and reporting, one always comes across producing categorical tables used for analysis. Provided with lots of different kind of data types such as labs, adverse events, physical examination, concomitant therapies etc and multiple categories for several categorical variables, it becomes difficult sometimes to impute 0 values for categories which are missing altogether for a particular variable. SAS provides multiple options to handle such cases and produce output with all the categories even if some don’t exist in the data. This paper concentrates on different work around methods and will discuss details of implementing each technique.

INTRODUCTION

During Analysis and reporting in SAS, it is common practice to summarize all possible combinations of class level variables even if few categories are altogether missing in the database. As for instance, Case Report Form has an option as ‘OTHER’ in Race category but none of the subjects qualifies for it in current data release and thus race category ‘OTHER’ is missed in the database. But in the above scenario, report still needs to summarize ‘OTHER’ category under race with zero value assuming this category might appear in future data views. There are few techniques which we will discuss in this paper to summarize all the categories in the report and fill with zero those categories which are missing from the data.

We will use HDL Lab data as shown in Figure 1 for illustration purpose throughout this compilation. We want to create a descriptive summary report of this data categorized by sex and abnormality ranges. This HDL Lab data can be categorized into ‘Low’, ‘Normal’ or ‘High’ categories based on following criteria.

<table>
<thead>
<tr>
<th>HDL (Male)</th>
<th>&lt; 25 mg/dL</th>
<th>Low</th>
</tr>
</thead>
<tbody>
<tr>
<td>25 mg/dL - 65 mg/dL</td>
<td>Normal</td>
<td></td>
</tr>
<tr>
<td>&gt; 65 mg/dL</td>
<td>High</td>
<td></td>
</tr>
<tr>
<td>HDL (Female)</td>
<td>&lt; 38 mg/dL</td>
<td>Low</td>
</tr>
<tr>
<td>38 mg/dL – 94 mg/dL</td>
<td>Normal</td>
<td></td>
</tr>
<tr>
<td>&gt; 94 mg/dL</td>
<td>High</td>
<td></td>
</tr>
</tbody>
</table>

FIGURE 1: HDL Lab Data
Following user defined formats will also be used in all the examples explained in this paper.

```
Proc Format;
  Value range 1='Low'
                2='Normal'
                3='High'
  Value sex 1 = 'Male'
                 2 = 'Female'
Run;
```

**METHOD 1: USE OF DUMMY DATASET**

A simple method to achieve our purpose can be to create a dummy hard coded dataset with all the categories that we expect in the report. Get the output dataset by running any statistical procedure on original data and merge this output dataset with previously created dummy dataset. Following piece of code explains this in detail.

```
/* Hard coded dummy dataset in the format same as expected in final output */
Data Dummy(drop = i j);
  Do i = 1 to 3;
    Do j = 1 to 2;
      SEX = j;
      RANGE = i;
      N = 0;
      MEAN = .;
      MEDIAN = .;
      STD = .;
      MIN = .;
      MAX = .;
      Output;
    End;
  End;
Run;

/* Create a dataset containing summary statistics */
Proc Means Data = HDL noprint;
  Class SEX RANGE;
  Var LBRSLT;
  Output out = Stat(keep = SEX RANGE N Mean Median Std Min Max)
           N = N
           Mean = Mean
           Median = Median
           Std = Std
           Min = Min
           Max = Max;
Run;

/* Sort Stat and Dummy dataset by SEX and RANGE and Merge them together */
Proc Sort Data = Stat;
  By SEX RANGE;
  Where SEX > . and RANGE > .;
Run;
Proc Sort Data = Dummy;
  By SEX RANGE;
Run;
```
Data Stat;
    Merge Dummy Stat;
    By SEX RANGE;
    Format SEX sex. RANGE range.; /* Apply previously defined formats */
Run;
Proc Print noobs;
Run;

The above code will create the output as shown in Figure 2.

<table>
<thead>
<tr>
<th>SEX</th>
<th>RANGE</th>
<th>N</th>
<th>MEAN</th>
<th>MEDIAN</th>
<th>STD</th>
<th>MIN</th>
<th>MAX</th>
</tr>
</thead>
<tbody>
<tr>
<td>Male</td>
<td>Low</td>
<td>0</td>
<td>42.3333</td>
<td>42.0</td>
<td>3.51188</td>
<td>.</td>
<td>.</td>
</tr>
<tr>
<td>Male</td>
<td>Normal</td>
<td>3</td>
<td>92.5000</td>
<td>92.5</td>
<td>2.12132</td>
<td>39</td>
<td>46</td>
</tr>
<tr>
<td>Male</td>
<td>High</td>
<td>2</td>
<td>30.6667</td>
<td>31.0</td>
<td>2.51661</td>
<td>28</td>
<td>33</td>
</tr>
<tr>
<td>Female</td>
<td>Low</td>
<td>3</td>
<td>96.0000</td>
<td>96.0</td>
<td>1.41421</td>
<td>95</td>
<td>97</td>
</tr>
<tr>
<td>Female</td>
<td>Normal</td>
<td>0</td>
<td>.</td>
<td>.</td>
<td>.</td>
<td>.</td>
<td>.</td>
</tr>
<tr>
<td>Female</td>
<td>High</td>
<td>2</td>
<td>.</td>
<td>.</td>
<td>.</td>
<td>.</td>
<td>.</td>
</tr>
</tbody>
</table>

FIGURE 2: Output from METHOD 1

METHOD 2: PROC MEANS USING ‘COMPLETETYPES’ AND ‘PRELOADFMT’ OPTION

The COMPLETETYPES option available in PROC MEANS creates all possible combinations of class variables even if combination doesn’t exist in the input dataset. If this option is used in conjunction with PRELOADFMT option in CLASS statement, PROC MEANS will write all user-defined format ranges or values for the combinations of class variables to the output, even when a frequency is zero.

Following code will demonstrate the use of these two options. We will use the same dataset from Figure 1 and previously created user-defined formats RANGE and SEX for this illustration.

Proc Means Data = HDL noprint completetypes;
    Format SEX sex. RANGE range.;
    Class SEX RANGE/preloadfmt;
    Var LBRSLT;
    Output out = Stat(keep = SEX RANGE N Mean Median Std Min Max)
                N = N
                Mean = Mean
                Median = Median
                Std = Std
                Min = Min
                Max = Max ;
Run;
Proc Print noobs;
    Where SEX > . and RANGE > .;
Run;

The above code will create the output as shown in Figure 3
It is important to note at this point that above code without PRELOADFMT option will work same as SPARSE option in Proc Freq.

**METHOD 3: PROC TABULATE USING ‘PRINTMISS’ AND ‘PRELOADFMT’ OPTION**

The PRINTMISS option in TABLE statement of PROC TABULATE enables you to display all the possible combinations of CLASS variable values that exist in the data set. Thus, if the CLASS variable value is present in one crossing but not another, all you may need to add is the PRINTMISS option to the TABLE statement. Use of PRELOADFMT option in conjunction with PRINTMISS option works the same way as explained above in METHOD 2.

Following piece of code explains the use of these two options in Proc Tabulate.

```
Proc Tabulate Data = HDL;
  Format SEX sex. RANGE range.;
  Class SEX RANGE/preloadfmt;
  Var LBRSLT;
  Table SEX='SEX'*RANGE='RANGE',
      LBRSLT='LBRSLT' N='N'
      LBRSLT='LBRSLT' MEAN='MEAN'
      LBRSLT='LBRSLT' MEDIAN='MEDIAN'
      LBRSLT='LBRSLT' STD='STD'
      LBRSLT='LBRSLT' MIN='MIN'
      LBRSLT='LBRSLT' MAX='MAX' / printmiss;
Run;
```

The above code will create the output as show in Figure 4

```
<table>
<thead>
<tr>
<th>SEX</th>
<th>RANGE</th>
<th>N</th>
<th>MEAN</th>
<th>MEDIAN</th>
<th>STD</th>
<th>MIN</th>
<th>MAX</th>
</tr>
</thead>
<tbody>
<tr>
<td>Male</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>Low</td>
<td></td>
<td>0</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>Normal</td>
<td></td>
<td>3</td>
<td>42.3333</td>
<td>42.0000</td>
<td>3.51188</td>
<td>39</td>
<td>46</td>
</tr>
<tr>
<td>High</td>
<td></td>
<td>2</td>
<td>92.5000</td>
<td>92.5000</td>
<td>2.1232</td>
<td>91</td>
<td>94</td>
</tr>
<tr>
<td>Female</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>Low</td>
<td></td>
<td>3</td>
<td>30.6667</td>
<td>31.0000</td>
<td>2.51661</td>
<td>20</td>
<td>33</td>
</tr>
<tr>
<td>Normal</td>
<td></td>
<td>0</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>High</td>
<td></td>
<td>2</td>
<td>96.0000</td>
<td>96.0000</td>
<td>1.41421</td>
<td>95</td>
<td>97</td>
</tr>
</tbody>
</table>
```

**METHOD 4: PROC REPORT USING ‘COMPLETEROWS’ AND ‘PRELOADFMT’ OPTION**

Using PRELOADFMT option in conjunction with COMPLETEROWS option in Proc Report outputs all the combinations of category variables even if they are absent in the dataset. Following code demonstrates how to calculate summary statistics using these two options in Proc Report.
Since Proc Report will not allow calculating and displaying different summary statistics for same variable (LBRSLT in our case) in the output, we will create dummy variables (one for each required statistic) first in the input dataset.

```sas
/* Create Dummy variables for each expected statistics */
Data HDL1;
  Set HDL;
  N=LBRSLT;
  MEAN=LBRSLT;
  MEDIAN=LBRSLT;
  STD=LBRSLT;
  MIN=LBRSLT;
  MAX=LBRSLT;
Run;

Proc Report Data=hdl1 completerows nowd BOX;
  Format SEX sex. RANGE range. ;
  Column SEX RANGE N MEAN MEDIAN STD MIN MAX;
  Define SEX / order = internal group preloadfmt;
  Define RANGE / order = internal 'RANGE' group preloadfmt;
  Define N / analysis n;
  Define MEAN / analysis mean;
  Define MEDIAN / analysis median;
  Define STD / analysis std;
  Define MIN / analysis min;
  Define MAX / analysis max;
Run;
```

The above code will create the output as show in Figure 5

<table>
<thead>
<tr>
<th>SEX</th>
<th>RANGE</th>
<th>N</th>
<th>MEAN</th>
<th>MEDIAN</th>
<th>STD</th>
<th>MIN</th>
<th>MAX</th>
</tr>
</thead>
<tbody>
<tr>
<td>Male</td>
<td>Low</td>
<td>0</td>
<td>.</td>
<td>.</td>
<td>.</td>
<td>.</td>
<td>.</td>
</tr>
<tr>
<td></td>
<td>Normal</td>
<td>3</td>
<td>42.333333</td>
<td>42</td>
<td>3.5118845</td>
<td>39</td>
<td>46</td>
</tr>
<tr>
<td></td>
<td>High</td>
<td>2</td>
<td>92.5</td>
<td>92.5</td>
<td>2.1213203</td>
<td>91</td>
<td>94</td>
</tr>
<tr>
<td>Female</td>
<td>Low</td>
<td>3</td>
<td>30.666667</td>
<td>31</td>
<td>2.516615</td>
<td>28</td>
<td>33</td>
</tr>
<tr>
<td></td>
<td>Normal</td>
<td>0</td>
<td>.</td>
<td>.</td>
<td>.</td>
<td>.</td>
<td>.</td>
</tr>
<tr>
<td></td>
<td>High</td>
<td>2</td>
<td>96</td>
<td>96</td>
<td>1.4142135</td>
<td>95</td>
<td>97</td>
</tr>
</tbody>
</table>

**FIGURE 5: Output from METHOD 4**

**CONCLUSION**

When all the expected data categories are not available in the data, but still need to be presented in the categorical summary reports, PRELOADFMT option comes very handy. To generate custom reports that also include all missing categorical combinations, usually programmer has to go through lot of data manipulation and many times programmer has to do some hard coding. Use of PRELOADFMT option in conjunction with other procedure specific options like COMPLETETYPES, COMPLETEROWS, and PRINTMISS reduces the hassle of data manipulations and it also takes care of user defined format for the categories in the report. This reduces programming efforts significantly and also keeps the entire reporting macro free of unnecessary steps that will come up otherwise.
ACKNOWLEDGEMENTS

We would like to acknowledge Sapan Raval for his help in putting this paper together.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the authors at:

Name: NIRAJ J PANDYA
Phone: 201-936-5826
E-mail: npandya@elementtechnologies.com

Name: RAMALAXMAREDDY KANDIMALLA
Phone: 408-207-2293
E-mail: krlaxmareddy@gmail.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.